

Capítulo 3

Funciones condicionales

Función SI

Función O

Función Y

Anidar la función SI

Función CONTAR.SI

Función SUMAR.SI

Ejercicios de autoevaluación

Formato condicional

Ejercicios de autoevaluación

Funciones de búsqueda y referencia

Nombres de bloque

Sintaxis de la función BUSCARV

Sintaxis de la función BUSCARH

Ejercicios de autoevaluación

Funciones financieras

Funciones para operaciones de capitalización o descuento y rentas Valor neto actual y tasa interna de retorno

Ejercicios de autoevaluación

Validación

Ejercicio de autoevaluación

Trabajo práctico 3

Funciones Condicionales

Una pequeña financiera necesita evaluar las solicitudes de crédito que realizan algunos de sus clientes, para lo cual utiliza una planilla de Microsoft Excel que guarda alguna información de interés para tomar estas decisiones.

Abra el archivo 0301.xls desde la carpeta Guías del CD.


La planilla que observaremos tiene ya varios datos ingresados, correspondiendo al nombre de los clientes en la columna A, la edad de cada uno en la columna siguiente, el historial de la persona en la columna C, en donde se indica si ha sido moroso en alguna oportunidad, su límite de crédito en la columna D y finalmente el monto solicitado en la columna E.

	A	B	C	D	E
4	NOMBRE	EDAD	HISTORIAL	LIMITE	SOLICITADO
5	LOPEZ SANDRA	18		2500	1025
6	BLAUCHET BELKIS	25		1800	2760
7	BORGHELLO OSCAR	54		3500	3420
8	FERNANDEZ VANESSA	36	Moroso	2000	2760
9	DECK PABLO	42	Moroso	1500	560
10	PUGLIOTTI PEDRO	35		2600	940
11	BAUZA MARTINA	28	Moroso	4000	3460
12	HUBER CARLOS	33		3000	2600
13	NEONIS SONIA	19		1000	1850
14	RACCATI MIGUEL	54		2500	2350

Función SI


Esta función nos permite evaluar una determinada condición, para obtener un resultado en función de que la misma sea o no verdadera. En el ejemplo que estamos analizando, podríamos evaluar si un cliente ha solicitado un crédito por un monto superior a su límite de crédito, lo cual expresariamos literalmente de la siguiente manera: "Si es verdad que el límite de crédito es mayor al monto solicitado, entonces esta habilitado, en caso contrario el crédito es denegado".

Gráficamente, este razonamiento se representaría de esta manera:




En la celda F4 se ha ingresado el rótulo CONDICIÓN, debajo de esta celda ingresaremos una función que permita a la planilla realizar la misma acción que expresamos en el razonamiento anterior.

Si utilizaremos el asistente de funciones (ver Apéndice, Funciones de Microsoft Excel, Asistente de funciones), eligiendo la función SI dentro de las funciones lógicas, obtendríamos el siguiente cuadro de dialogo que utilizaremos para analizar los parámetros de la función SI.



Como podemos observar, el primer parámetro que nos solicita es la Prueba_lógica que en nuestro ejemplo equivale a verificar si "el límite de crédito es mayor al monto solicitado"; sabemos que podría expresarse esto haciendo referencia a las celdas que contienen los valores, lo que para el primer cliente equivale a indicar: D5>E5, ya que en la columna D esta el límite de crédito y en la columna E el monto solicitado.

La prueba lógica debe entenderse como la expresión de una condición que puede ser evaluada y resultar verdadera o falsa. Será verdadera cuando se cumpla la condición y falsa en el caso contrario. Para poder expresar esta condición, debemos recurrir a los operadores lógicos que nos permitirán comparar entre dos términos.



El segundo argumento es Valor_si_verdadero, vale decir que aquí se debe indicar cual es la acción que debe tomar la aplicación en el caso de que se cumpla la condición de la prueba lógica. Aquí podemos ingresar un texto entre comillas si queremos que aparezca un mensaje, o un número o una operación si necesitamos valores. En nuestro caso, necesitamos que si se cumple la condición, o sea que el valor indicado en D5 sea mayor que el que aparece en E5, aparezca el texto "Habilitado".

Finalmente, el tercer parámetro es Valor_si falso, lo que supone que en él se indicara la acción que se debe realizar cuando la condición de la prueba lógica no se cumpla. Al igual que el caso anterior, podemos ingresar texto o números, pero aquí requerimos que cuando el valor de D5 no sea mayor al valor de E5, aparezca en la celda el texto "EXCEDIDO".

Ud. puede utilizar el Asistente de Funciones para ingresar esta función, haciendo clic en el botón Pegar función de la barra de herramientas y en Categoría de función Lógicas, elegir la función SI, o bien haga activa la celda F5 y escriba la función

`=SI(D5>E5;"Habilitado";"EXCEDIDO")`

Cuando presione Enter para aceptar esta función, en la celda F5 podrá ver que aparece la palabra "Habilitado", ya que se ha evaluado que para ese cliente el límite de crédito de la celda D5 es mayor al monto solicitado en la celda E5.

Copie la función recién ingresada en F5 al resto de las celdas de 1, columna. Puede hacer llenado automático si así lo prefiere. Observe que sucede en cada una de las celdas al terminar de copiar esta función.



A aquellas personas que han solicitado un importe mayor a su límite de crédito, o sea que no cumplen con la condición expresada en la prueba lógica, les aparece la palabra "Excedido" junto al monto solicitado.

Función O


La Financiera ha establecido en su normativa que las personas mayores de 50 años presentan un riesgo Alto en su capacidad de pago, lo cual también sucede en el caso de que tengan antecedentes de morosidad. Para eso ha previsto en la planilla la columna G, que en la celda G4 muestra el rótulo RIESGO.

El riesgo de cada cliente podrá ser Normal o Alto, dependiendo esto de dos condiciones: "Si la persona es mayor de 50 años o si tiene antecedentes de morosidad", puede ser calificada con riesgo Alto.

Esta claro que podemos evaluar si $B5 > 50$ y luego es posible verificar si $C5 = \text{"Moroso"}$, con lo cual se estarían anidando dos funciones condicionales, lo que gráficamente se representaría de esta manera:




Ahora bien, el problema es cómo lograr que Microsoft Excel evalúe ambas condiciones al mismo tiempo, con lo cual se simplificaría el esquema, pudiendo graficarse de este modo:



Debemos destacar que en este caso las normas de la Financiera establecen que basta con que una de estas dos condiciones se cumpla para que el cliente se considere de alto riesgo. Es decir que no deben cumplirse ambas condiciones, sino que será suficiente que se cumpla una o la otra.

Si utilizáramos el asistente de funciones como en el caso anterior, obtendríamos el siguiente cuadro de dialogo:



En este cuadro podemos observar que la función O resultara verdadera en caso de que alguno de sus argumentos sea verdadero y resultara falsa si ninguno de ellos se cumple. Ahora bien, los argumentos de esta función son dos o mas condiciones que se identifican aquí como Valor_lógico1, numerandose en orden correlativo hasta tantas condiciones como queramos utilizar, siempre que no superen las treinta.

Para utilizar esta función vinculando dos o más condiciones debemos indicar la función O y encerrar entre paréntesis las condiciones como parámetros, ingresando el separador correspondiente entre cada uno de los valores lógicos (ver Apéndices, Configuración Regional).

En nuestro caso si ingresáramos en la celda G5 la función tal como se ha explicado: =O(B5>50;C5="Moroso"), evaluando si se cumple alguna de las dos condiciones, obtendríamos la palabra FALSO, ya que ninguna de las dos condiciones se cumple. Si analizamos los datos de la planilla, en el caso del tercer y cuarto cliente, nos mostrara la palabra VERDADERO, ya que al menos una de las dos condiciones se cumple en cada caso.

Ahora bien, nosotros necesitamos conocer el riesgo de cada cliente, el que será Alto si se cumple alguna de las dos condiciones, y en caso contrario será Normal. Obviamente para ello debemos utilizar una función SI, la que evaluará si la prueba lógica es verdadera o falsa, tal como lo hemos visto anteriormente.

Lo que haremos para lograr nuestro objetivo es ANIDAR estas dos funciones, consiguiendo así que la función SI responda según se cumpla al menos una de las condiciones requeridas. Esto equivale a decir que en este caso la Prueba_lógica será precisamente la función O con sus correspondientes parámetros, mientras que los textos "Alto" y "Normal" corresponden a los parámetros Valor_si_verdadero y Valor_si_falso respectivamente.

Haga activa la celda G5 y escriba la función

=SI(O(B5>50;C5="Moroso");"Alto";"Normal")

Cuando presione Enter para aceptar esta función, en la celda G5 podrá ver que aparece la palabra "Normal", ya que se ha evaluado que para ese cliente la edad no es mayor a 50 y su historial no es igual a Moroso.

Copie la función recién ingresada en G5 al resto de las celdas de la columna. Puede hacer llenado automático si así lo prefiere. Observe qué sucede en cada una de las celdas al terminar de copiar esta función.



A aquellas personas que tienen más de 50 años o aparecen señalados como Morosos, con lo que se cumple con la condición expresada en la prueba lógica, les aparece la palabra "Alto".

Función Y

La Financiera ha establecido en su normativa que las solicitudes de crédito resultaran aprobadas o denegadas dependiendo de las evaluaciones anteriores. Para eso ha previsto en la planilla la columna H, que en la celda H4 muestra el rótulo DECISIÓN.

La decisión para cada cliente podrá ser Aprobado o Denegado, dependiendo esto de dos condiciones concurrentes, ya que "Si la condición es Habilitado Y su riesgo se califique como Normal", el crédito solicitado es Aprobado.

Esta claro que podemos evaluar si E5="Habilitado" y también si G5="Normal"; la cuestión es cómo lograr que Microsoft Excel evalúe que ambas condiciones se cumplan al mismo tiempo.

Debemos destacar que en este caso las normas de la Financiera establecen que AMBAS condiciones se deben cumplir para que el crédito se considere aprobado. Es decir que no es suficiente conque se cumpla una o la otra condición.

Si utilizáramos el asistente de funciones como en el caso anterior, obtendríamos un cuadro de dialogo similar

En el que podemos observar que la función Y resultara verdadera en caso de que sus argumentos sean verdaderos y resultara falsa si alguno de ellos no se cumple. Como en la función O, los argumentos de la función Y son dos o mas condiciones, que se identifican como Valor_lógico1 y se numeran en orden correlativo hasta tantas condiciones como queramos utilizar.

Para utilizar esta función vinculando dos o mas condiciones debemos indicar la función Y para encerrar entre paréntesis las condiciones como parámetros, ingresando el separador correspondiente entre cada uno de los valores lógicos (ver Apéndices, Configuración Regional).

En nuestro caso si ingresáramos en la celda H5 la función tal como se ha explicado: =Y(F5="Habilitado";G5="Normal"), evaluando si se cumplen las dos condiciones, obtendríamos la palabra VERDADERO, ya que ambas condiciones se cumplen. Si analizamos los datos de la planilla, en el caso de los siguientes tres clientes, nos mostrara la palabra FALSO, ya que al menos una de las dos condiciones no se cumple en cada caso.

Nosotros necesitamos conocer la Decisión para cada cliente, la que será Aprobado en el caso de que se cumplan ambas condiciones, y en caso contrario será Denegado. Obviamente, para ello debemos utilizar una función SI, la que evaluara si la prueba lógica es verdadera o falsa, tal como lo hemos visto anteriormente.

Lo que haremos para lograr nuestro objetivo, al igual que en el caso anterior, es ANIDAR estas dos funciones, consiguiendo así que la función SI responda si se cumplen las dos condiciones requeridas. Esto equivale a decir que en este caso la Prueba_lógica será precisamente la función Y, con sus correspondientes parámetros, mientras que los textos "Aprobado" y "DENEGADO" corresponden a los parámetros Valor_si_verdadero y Valor_si_falso respectivamente.

Haga activa la celda H5 y escriba la función

=SI(Y(F5="Habilitado";G5="Normal");"Aprobado";"DENEGADO")

Cuando presione Enter para aceptar esta función, en la celda H5 podrá ver que aparece la palabra "Aprobado", ya que se ha evaluado que para ese cliente, su Condición es Habilitado y su Riesgo es Normal.

Copie la función recién ingresada en H5 al resto de las celdas de la columna. Puede hacer llenado automático si así lo prefiere. Observe qué sucede en cada una de las celdas al terminar de copiar esta función.



A aquellas personas cuya Condición es Excedido o que su Riesgo es calificado como Alto, con lo que no cumplen con la condición expresada en la prueba lógica, les aparece la palabra "DENEGADO".

Anidar la función SI

La Financiera ha decidido no descartar en primera instancia a aquellas solicitudes que resulten denegadas, para lo cual ha definido una serie de criterios que permiten revisar estas decisiones, maximizando así el mantenimiento de los clientes. Para eso ha previsto en la planilla la columna I, que en la celda 14 muestra el rótulo VERIFICAR.

Analizando la situación se ha comprobado que algunas solicitudes denegadas lo han sido porque el monto solicitado es mayor al límite de crédito, aun cuando el cliente tiene una calificación de riesgo Normal. Tal es el caso del segundo cliente. Es claro que en este caso podría reconsiderarse la decisión y, sobre todo, analizar cuanto es la diferencia entre ambos valores.

El tercer cliente muestra otra situación, ya que si bien esta Habilitado, ha sido calificado con riesgo Alto en función de su edad. Es claro que en este caso también podría reconsiderarse la decisión buscando una solución alternativa.


El crédito del cuarto cliente ha sido denegado, pero en este caso la razón es que

ha sido Moroso anteriormente y en esta oportunidad esta solicitando un importe superior al de su límite de crédito.

El quinto cliente muestra otra situación, ya que si bien esta Habilitado, ha sido calificado con riesgo Alto en función de que aparece como Moroso en los registros. Es posible en este caso al menos verificar en su legajo las razones por las que quedó moroso.

En este caso tenemos para evaluar una condición bastante compleja, la que literalmente podemos expresar de la siguiente manera: "SI la Condición es Excedido y el Riesgo es Normal, entonces debe mostrar la diferencia entre el límite de crédito y el importe solicitado; en caso contrario, SI la Edad es mayor a 50, entonces debe solicitar un Garante; en caso contrario, SI el Historial es Moroso y la Condición es Habilitado, entonces debe revisarse el legajo, en caso contrario no se debe mostrar nada, ya que el crédito esta bien denegado.

Gráficamente, el complejo razonamiento anterior se representaría de este modo:



Por supuesto, podemos evaluar cada una de estas condiciones por separado; sin embargo, lo que aquí necesitamos es que una función SI tome como Valor_si_falso una nueva función SI.

En la celda 14 se ha ingresado el rótulo VERIFICAR, debajo de esta celda ingresaremos una función que permita a la planilla realizar la misma acción que expresamos en el razonamiento anterior.

Haga activa la celda 15 y escriba la función

=SI(Y(F7="EXCEDIDO";G7="Normal");E7-D7;SI(B7>50;"Pedir GARANTE";SI(Y(C7="Moroso";F7="Habilitado");"Ver legajo";""))))

Cuando presione Enter para aceptar esta función, en la celda I5 podrá ver que no aparece nada, ya que se ha evaluado que para ese cliente, su Condición NO es Excedido y su Riesgo es Normal, su edad NO es mayor a 50 y su Historial NO es Moroso y su Condición es Habilitado.

Copie la función recién ingresada en 15 al resto de las celdas de la columna. Puede hacer llenado automático si así lo prefiere. Observe qué sucede en cada una de las celdas al terminar de copiar esta función.




A aquellas personas cuya Condición es Excedido y que su Condición es calificada como Normal, al cumplir con la condición expresada en la primera prueba lógica, les aparece la diferencia entre lo solicitado y su límite de crédito; para los que no cumplen lo anterior y su Edad es mayor a 50, se les solicita un Garante; a los que no cumplen con todo lo anterior y su Historial es Moroso y su Condición es Habilitado, se les sugiere revisar su Legajo; finalmente al resto de los clientes no se les muestra texto ni valor alguno.

Función CONTAR.SI

La Financiera necesita conocer la cantidad de créditos aprobados en primera instancia. Para eso ha previsto en la planilla la fila 37, que en la celda A37 muestra el rótulo CREDITOS OTORGADOS.

Normalmente las fórmulas realizan cálculos con todos los valores de un rango determinado, por lo que en este caso resulta algo difícil contar cuantos créditos fueron otorgados ya que en el rango H5:H34 hay celdas que muestran el texto "Aprobado" y otras que dicen "DENEGADO". Está claro que si resolviéramos manualmente esta cuestión, lo que haríamos es contar aquellas celdas que cumplen con una condición, vale decir que las vamos a CONTAR SI es verdadera la condición y en caso contrario no las tendremos en cuenta.

Si utilizáramos el asistente de funciones para generar esta función, obtendríamos el siguiente cuadro de dialogo:



El parámetro Rango nos remite al conjunto de celdas que queremos contar.

El argumento Criterio nos permite incorporar un valor que queremos comparar con la celda para que esta sea considerada valida.

En nuestro caso el rango a considerar es 115:1134 y dentro de ese bloque queremos contar solamente aquellas celdas que tengan el texto "Aprobado".

Haga activa la celda B37 y escriba la función

`=CONTAR.SI(H5:H34;"Aprobado")`

Cuando presione Enter para aceptar esta función, en la celda B37 podrá ver que se muestra el número 14, lo cual corresponde a la cantidad de créditos aprobados.


Función SUMAR.SI

Por supuesto también se necesita conocer el importe total en concepto de créditos aprobados en primera instancia. Para eso ha previsto en la planilla la fila 38, que en la celda A38 muestra el rótulo TOTAL ACUMULADO.

Si resolviéramos manualmente esta cuestión, lo que haríamos es sumar el importe solicitado de los clientes cuyos créditos hayan sido aprobados, vale decir que las vamos a SUMAR SI es verdadera la condición y en caso contrario no las tendremos en cuenta.

Obviamente la diferencia aquí es que no estamos contando sino sumando valores, y que la selección de estos se hace en función de una condición verificada en otras celdas.

Si utilizáramos el asistente de funciones para generar esta función, obtendríamos el siguiente cuadro de dialogo:



El parámetro Rango nos remite al conjunto de celdas que contienen la condición a evaluar, en nuestro caso la Decisión.

El argumento Criterio nos permite incorporar un valor que queremos comparar con la celda para que sea considerada valida; en nuestro caso, si es Aprobado. Rango_suma corresponde al conjunto de celdas que contienen los valores que queremos sumar cuando se cumpla la condición del Criterio.

En el ejemplo, el rango a considerar es H5:H34 y del bloque E5:E34 queremos sumar solamente aquellas celdas que tengan el texto "Aprobado" en la columna H.

Haga activa la celda B38 y escriba la función

=SUMAR.SI(H6:H34;"Aprobado";E5:E34)

Cuando presione Enter para aceptar esta función, en la celda B38 podrá ver que se muestra el monto total de los créditos otorgados, lo cual corresponde a la suma únicamente de aquellos créditos que en la columna H están identificados con el texto aprobados.

Ejercicios de Autoevaluación

Ejercicio 1

En la columna J, se necesita una función que evalúe si el monto que se ha excedido (mostrado en la columna 1) es inferior al 40% del límite de crédito, ya que en este caso debe mostrarse la palabra "Posible".



Ejercicio 2

En la columna K, escriba una función que indique con la palabra "Premium" a los clientes que tienen un límite de crédito mayor a 3.000. A los que no cumplen con esa condición se les mostrara solamente un guión: "-".



Ejercicio 3

En la columna L, escriba una función para que a aquellos clientes menores de 40 años y que no sean morosos se les identifique con la palabra "VIP", ya que forman la cartera de clientes que se deberá atender de forma especial.



Ejercicio 4

En la celda B39 ingrese una función para contar los créditos que han sido denegados, y en la celda B40 escriba una función que sume los montos de los créditos que no han sido aprobados.



Formato Condicional

El Formato Condicional es un tipo de formato específico que nos resulta especialmente útil al momento de querer asignar a una celda o a un rango de celdas determinadas características específicas de formato (tipo de fuente, tamaño, color, bordes, etc.), en función de los distintos contenidos que tenga la celda.

Para poder ver cómo funciona este formato condicional tomaremos un ejemplo.

Abra el archivo 0302.xls del CD, carpeta Guías, y encontrará una tabla de estas características:

	A	B	C	D	E	F
1		RESUMEN MENSUAL: CUENTAS CORRIENTES				
2	CLIENTE	COMPRA	PAGOS	SALDO	OBSERVAC.	
3	ARIAS, Carlo	\$ 138,98	\$ 50,00	\$ 88,98	Saldo Deudor	
4	ALVAREZ, M	\$ 475,32	\$ 250,00	\$ 225,32	Saldo Deudor	
5	BORSOTTI, E	\$ 546,26	\$ 326,00	\$ 220,26	Saldo Deudor	
6	BRUNO, Amé	\$ 548,00	\$ 456,00	\$ 92,00	Saldo Deudor	
7	CORNAGLIA	\$ 325,00	\$ 400,00	\$ (75,00)	Saldo acreedor	
8	CRAVERO, G	\$ 125,00	\$ 200,00	\$ (75,00)	Saldo acreedor	
9	DIAZ, Eugen	\$ 145,30	\$ 100,00	\$ 45,30	Saldo Deudor	
10	ERMAN, Nori	\$ 254,10	\$ 100,00	\$ 154,10	Saldo Deudor	
11	FABRI, Artur	\$ 850,00	\$ 60,00	\$ 790,00	Saldo Deudor	
12	FEDELE, Jav	\$ 156,00	\$ 156,00	\$ -	Saldo acreedor	
13	GONZALEZ,	\$ 321,30	\$ 300,00	\$ 21,30	Saldo Deudor	
14	IMBACH, Om	\$ 457,90	\$ -	\$ 457,90	Saldo Deudor	
15	SARSOTTI, N	\$ 215,65	\$ 20,00	\$ 195,65	Saldo Deudor	

Se puede apreciar que se trata de un resumen de cuentas mensual, que elabora una empresa con las compras totales de sus clientes, los pagos realizados y el saldo correspondiente.

Las columnas compras y pagos son datos ingresados manualmente, en cambio el saldo es una fórmula que consiste en la diferencia de lo comprado y lo pagado. El saldo del primer cliente (celda E8) es la fórmula =C8-D8. Del mismo modo las restantes celdas de esta columna son una copia de esta fórmula, que por el hecho de tener referencias relativas, irán incrementando respectivamente la coordenada fila (=C9-D9; =C 10-D10; =C 11 -D 11; etc.).

Si observa el caso de los clientes Cornaglia y Cravero, ambos poseen un saldo negativo que aparecen entre paréntesis, lo que en realidad indica que tienen un

saldo favorable a ellos (saldo acreedor).


Sería interesante que indique en distintos colores cuando el saldo es deudor o acreedor.

Para ello, primero seleccione el rango de saldos F8:F20, y a continuación elija en el menú Formato > Formato condicional.

Observamos el cuadro de dialogo que se muestra en la página siguiente.

En la primera lista aparece por defecto la opción Valor de la celda (otra opción a elegir en esta lista es Fórmula).

Elija Valor de la celda.



En este caso queremos que las celdas que tengan un valor mayor que 0, por ser un saldo deudor, aparezcan en rojo.

En la segunda lista desplegable, en lugar de la opción por defecto Entre, seleccione Mayor que, y en la tercera lista escriba 0. 

Ya tenemos la condición definida (Valor de celda > 0), ahora haga clic en el botón Formato, y en el cuadro que aparece debemos cambiar la opción de color de la fuente por rojo. Luego haga clic en Aceptar, y nuevamente haga clic en Aceptar para regresar a la hoja. 

También puede agregar otras condiciones al formato condicional. Por ejemplo que en el caso de tener un saldo igual a cero, me muestre en color verde y en caso de ser menor que cero, me lo muestre en color azul.

Repita el mismo proceso anterior de seleccionar el rango de saldos y elegir Formato > Formato condicional. Ahora con el botón de Agregar me permite elegir las nuevas condiciones. 

Podemos ver que dentro de las características de formato que pueden manejarse con formato condicional, además del color de fuente que vimos en el demo, pueden modificarse tipo y tamaño de fuente, bordes y tramas.

Ejercicio de autoevaluación

Ejercicio 1:

En la columna Observaciones del libro 0302.xls, existe una función condicional SI que muestra una leyenda que indica Saldo Deudor o Saldo Acreedor según corresponda.

Utilice formato condicional en esta columna para modificar el color que mostrará esta leyenda, de tal modo que cuando corresponda Saldo Acreedor lo muestre en verde, y cuando corresponda Saldo Deudor lo muestre en rojo.



Funciones de Búsqueda y Referencia

Una librería administra el stock con un libro de Microsoft Excel y ha decidido diseñar una planilla con las compras programadas para una determinada fecha, que permita ver un detalle de las mismas así como el importe total a pagar por dicha compra.

Abra el archivo 0303.xls de la corneta Guías del CD-ROM

	A	B	C	D	E
1					
2					
3					
4	COMPRAS PROGRAMADAS PARA LA FECHA:				
5					
	COD-LIBRO	TITULO	CANTIDAD A COMPRAR	PRECIO COSTO	IMPORTE
6					
7					
8					
9					
10					
11					
12					
13					
14					
15					
16					
17	TOTAL IMPORTE				
18	BONIFICACION				
19	IMPORTE A PAGAR				

En la Hoja 1, el usuario deberá ingresar en la primera columna los códigos de libros que se desean adquirir; partiendo de este dato obtendremos de la Hoja 2 del libro el título del libro y el precio de costo. Cantidad a comprar es un dato que deberá ingresar el usuario, el importe se calcula con los datos de precio de costo y cantidad a comprar.

En la Hoja 2 encontramos una matriz de datos a la que denominaremos tabla compuesta por columnas identificadas por títulos de columna y filas que almacenan datos referidos a los libros que posee la librería en stock. Cada columna contiene datos similares, por ej. la columna título almacenara nombres de los libros y no otra cosa, la columna costo almacenara el precio de costo del libro y no otra cosa.

En la Hoja 3 encontraremos tablas con datos referidos a editoriales y porcentajes de bonificación que utilizamos para completar nuestra planilla.

En la celda E4 de la Hoja 1 ingresaremos la fecha de compra utilizando para ello una función de Microsoft Excel. Escriba en la celda F4 la función =Hoy()

En la celda A7 ingrese el código de libro 14104. En la celda B7 debe aparecer el título del libro correspondiente a dicho código, que según los datos de la Hoja 2 es "AL FILO DE LA DUDA".

	A	B
6	COD-LIBRO	TITULO
7	14104	AL FILO DE LA DUDA

¿Cómo mostramos en la celda B7 el nombre correspondiente al código de celda A7?


Una solución sería buscar el nombre correspondiente al código del libro 14104 (celda A7) en la Hoja 2 y escribirlo en la celda B7 de la Hoja 1. La desventaja principal de esta solución es la falta de automatización de la planilla (ver nota) ya que si tenemos que buscar el código 14104 entre 100.000 códigos diferentes demoraríamos mucho tiempo y existe la posibilidad de no encontrarlo; como desventajas secundarias podemos mencionar que la forma de escritura será siempre diferente y pueden existir errores ortográficos al no escribir exactamente el mismo nombre.

Otra solución sería buscar el nombre correspondiente al código del libro 14104 (celda A7) en la Hoja 2, copiarlo y pegarlo en la celda B7 de la Hoja 1. Nos encontraríamos con un problema similar al anterior.

Nota: La falta de automatización de la planilla se puede ver en el mismo momento que decidimos cambiar uno de los códigos de libros que pensábamos comprar. Imaginemos, por ejemplo, que no vamos a comprar el libro cuyo código es 14104 pero sí incorporamos en las compras programadas el código de libro 12050. De este modo tendremos que revisar otra vez todos los códigos de la Hoja 2 para encontrar el nombre correspondiente al código 12050.

Si tuviéramos que describir el razonamiento llevado a cabo para encontrar el nombre del libro correspondiente al código 14104, diríamos que realizamos una búsqueda en la columna A de la Hoja2 hasta encontrar dicho código y copiamos el nombre del libro que se encuentra 1° posición a la derecha, en la columna B.


La solución definitiva a la pregunta planteada unos párrafos atrás consiste en que Excel diseñe una función que automatice esa búsqueda, realizada por nosotros manualmente, y que nos permita encontrar valores a partir de un dato conocido.



La fórmula diseñada nos permitirá partir de un valor conocido (A7) al cual Excel denomina Valor buscado, para ingresar en un bloque de datos (Hoja2!A2:G114) al cual Excel denomina Matriz_buscar_en y obtener algún dato correspondiente al número de columna (2) deseado y al cual Excel denomina Indicador_columnas, luego de haber logrado la coincidencia entre el valor buscado y alguno de los valores de la columna 1 de la tabla.

Utilizaremos el Asistente de Funciones (Ver Capítulo 1, Asistente de Funciones) para que Microsoft Excel diseñe una fórmula de búsqueda.

Ubíquese en la celda B7 y acceda al botón de Pegar Función de la barra de herramientas. Seleccione la categoría de funciones de Búsqueda y referencia y dentro de ella la función BUSCARV



En el parámetro Valor_buscado escriba A7.

Acceda al botón contraer diálogo del parámetro Matriz_buscar_en y seleccione el bloque A2: G1 14 de la Hoja 2; finalice la operación haciendo clic nuevamente en el botón contraer diálogo y retornará al cuadro de diálogo de la función BUSCARV

En el parámetro Indicador_columnas coloque el numero 2.



En el parámetro ordenado escriba FALSO.

Haga clic en el botón Aceptar.

Nombres de bloque


Si analizamos el parametro Matriz_buscar_en podremos observar que al copiar la fórmula hacia abajo se modificaran los indicadores de fila. Para evitar este problema y para facilitar el manejo de un bloque de datos tan extenso que debemos seleccionar cada vez que utilizamos funciones de búsqueda, utilizaremos nombres de bloque.

Para trabajar con nombres de bloques utilizaremos el menú Insertar, opción Nombre. Aquí encontraremos todas las opciones que nos permiten crear, modificar y eliminar un nombre.

Acceda al menú Insertar/Nombre/Definir.

En la pantalla se abrirá el cuadro de dialogo que le permite crear un nuevo nombre indicando a continuación a qué bloque se esta refiriendo.


Escriba LIBROS como nombre de bloque y acceda al botón contraer diálogo para indicar el bloque A2:G114 de la Hoja 2.



Haga clic nuevamente en el botón contraer diálogo para regresar al cuadro de diálogo Definir nombre. Finalice presionando Aceptar.



Podemos verificar que el nombre esta correctamente creado desde el mismo menú insertar o accediendo al cuadro de nombres y seleccionando la referencia libros.



De aquí en mas cada vez que necesitemos trabajar con el bloque de datos A2:G114 de la Hoja 2 haremos referencia a LIBROS.

Si en algún momento necesitamos redefinir el bloque ya sea por haber incorporado más libros al final del bloque o porque se agregaran mas columnas con datos de los libros, lo único que debemos hacer es acceder al menú Nombre/Insertar y en al cuadro de dialogo anteriormente expuesto modificar el bloque con el botón de selección.

Entre las ventajas de trabajar con nombres de bloques podemos mencionar: evitar errores al copiar fórmulas ya que maneja referencias absolutas a celdas; facilitar la escritura de funciones cada vez mas complejas y con mas parámetros, y realizar el seguimiento del origen de los datos ante un posible error en el resultado de fórmulas y/o funciones.

Modificaremos ahora nuestra fórmula, reemplazando el segundo parámetro, Hoja!A2:G114 por el nombre libros. Acceda a la fórmula de la celda B7, modifíquela y verifique que quede como: =BUSCARV(A7,LIBROS,2,falso).

Nota: En los nombres de funciones así como en los nombres de bloque y referencias a celdas, el texto escrito en mayúsculas y minúsculas es equivalente.

La función BUSCARV busca un dato en una tabla, en las columnas a la derecha de aquella que contiene el valor buscado (código de búsqueda). Esta columna debe ser la primera de la Matriz_buscar_en ya que sólo podemos buscar en las columnas a la derecha de la misma. Si en nuestra tabla al campo CODIGO estuviera ubicado como tercera columna, debiéramos mover la misma hasta una posición que quede a la izquierda de aquella que contiene los datos que se desean buscar, es decir como primera columna.

Veamos algunos ejemplos.

Valor_Buscado	Matriz_Buscar_en	Indicador_Columnas	Resultado
A9=10316	LIBROS=A2:G114	2	COCINA: INDUSTRIA CASERA DEL QUESO.
A9=11417	LIBROS=A2:G114	4	MARTÍNEZ RICARDO O.
A9=12354	LIBROS=A2:G114	5	10,40
A9=13533	LIBROS=A2:G114	6	12,00

Sintaxis de la función BUSCARV

=BUSCARV(Valor_buscado, Matriz_buscar_en, Indicador_columnas, ordenado)

Valor_buscado: es el dato conocido y el que se busca en la primera columna de la Matriz_buscar_en. Puede ser un valor, una fecha, una cadena de texto o una referencia a celda.

Matriz_buscar_en: en el bloque de celdas donde se buscan los datos utilizando para la comparación la primera columna del bloque. Es conveniente darle un nombre al bloque como por ejemplo Base de datos, Lista, etc.

Indicador_columnas: es el número de columna que corresponde al dato que estamos buscando en Matriz_buscar_en la tabla numerándolas desde 1 para la columna mas a la izquierda.

ordenado: si la búsqueda que deseamos realizar es exacta, este parámetro debe ser FALSO. Si el bloque Matriz_buscar_en esta ordenado en forma ascendente por la primera columna, este parámetro será VERDADERO (expresión que puede omitirse de indicar), de lo contrario también será FALSO, expresión que debe escribirse como argumento de la función.

Notas: Si el argumento ordenado es VERDADERO se podrá omitir, y por lo tanto los valores de la primera columna de la Matriz_buscar_en deberán estar ordenados en forma ascendente. De lo contrario, BUSCARV podía devolver un valor incorrecto referido a búsquedas no exactas. Los valores de la primera columna de tabla de comparación pueden ser textos, números, valores lógicos o fechas. El texto escrito en mayúsculas y minúsculas es equivalente.

Continuamos con el ejemplo planteado, vamos a copiar la fórmula de la celda B7 para el resto de las filas hasta la celda B16. Puede realizar este procedimiento utilizando el llenado de fórmulas o Copiar y Pegar.

Aparece en el bloque B8:B16 un texto #N/A que significa que Excel esta devolviendo como resultado de la fórmula un valor de error ya que no puede resolverla adecuadamente.

El valor de error #N/A se da cuando un valor no esta disponible para una función o una fórmula. En nuestro ejemplo se presenta un valor inadecuado para Valor_buscado ya que las celdas A8:A16 están vacías.

Ingrese los siguientes códigos de libro en la columna A →

Celda	Valor
A8	10330
A9	11782
A10	12398
A11	13533
A12	12260
A13	10950
A14	11237
A15	14103
A16	11913

Observe cómo a medida que ingresamos los códigos aparece en las celdas de la columna B el título correspondiente a cada código.



Analicemos las fórmulas diseñadas por Excel en el bloque B8:B 16.

Celda	Valor_Buscado	Matriz_buscar_en	Indicador_Columnas	Celda	Resultado
A8	10330	LIBROS	2	B8	COCINA: PARA SENTIRSE BIEN
A9	11782	LIBROS	2	B9	RUEDAS (PK)
A10	12398	LIBROS	2	B10	SOBRE LA MUERTE Y LOS MORIBUNDOS
A11	13533	LIBROS	2	B11	NIÑOS DE NOCHE
A12	12260	LIBROS	2	B12	EDAD DE ORO DE LA CIENCIA FICCION, LA
A13	10950	LIBROS	2	B13	FRUTICULTURA
A14	11237	LIBROS	2	B14	PLANTAS DE INTERIOR
A15	14103	LIBROS	2	B15	GENERACION X
A16	11913	LIBROS	2	B16	CRIA DE LA CODORNIZ

Complete la columna cantidad ingresando por teclado los siguientes datos en la columna C.

En la celda D7 escriba una función BUSCARV que le permita obtener el precio de costo de los libros a comprar.



Copie esa fórmula para el resto de las filas.

En la celda E7 calcule el importe multiplicando la cantidad a comprar (celda C7) por el precio de costo (celda D7). Copie esa fórmula para el resto de las filas.

En la celda E17 debe escribir una función que le permita totalizar la compra del día. Utilice la función Suma (Ver Capítulo 1, Función Suma).

Celda	Cantidad
C7	4
C8	3
C9	5
C10	3
C11	2
C12	3
C13	4
C14	5
C15	4
C16	3

En la fila 18 la planilla debe indicar el porcentaje de bonificación que corresponda según el importe total de la compra. Estos porcentajes están tabulados en la hoja 3 del libro y varían según el importe total de la compra (celda E17) entre 0% y 10%.



	A	B	C	D	E	F	G	H
1	TABLA DE BONIFICACIONES							
2	IMPORTE	0,00	150,00	300,00	500,00	700,00	800,00	1.000,00
3	% DE BONIFICACION	0,00%	2,00%	3,50%	5,00%	7,50%	9,00%	10,00%

Podemos interpretar estas bonificaciones diciendo que si el importe total de compra es menor a \$ 150,00 no tendremos bonificación, si el importe total de compra es mayor o igual a \$150,00 y menor a \$300,00 tendremos un 2% de bonificación, si el importe total de compra se encuentra entre \$ 300,00 y \$ 500,00 (sin incluirlo) la bonificación será del 3,50 % y así para el resto de los importes considerando como porcentaje máximo de bonificación un 10 % para aquellos importes mayores o iguales que \$ 1.000,00.

Para mostrar el porcentaje de bonificación en la celda D18 utilizaremos una función de búsqueda en una tabla pero esta vez será en forma horizontal.

Ubíquese en la Hoja 3 y al bloque B2:H3 colóquele como nombre PORCENTAJES (ver en esta guía Nombres de Bloque).

En la Hoja 1 celda D18 escribiremos la función, en vez de diseñarla con el asistente de funciones como lo hicimos para la columna de título de libro.

Ubíquese el la celda D18 y escriba:

=BUSCARH(E17,PORCENTAJES,2).



	A	B	C	D	E	F
17		TOTAL IMPORTE			263,25	
18		BONIFICACION		=BUSCARH(E17;PORCENTAJES;2)		
19		IMPORTE A PAGAR				

La función de la celda D18 tiene idéntica cantidad de parámetros que la función BUSCARV explicada anteriormente. Su nombre cambia la letra V (vertical) por H (horizontal) debido a que los datos de la Matriz_buscar_en se están ubicando en forma horizontal.

En la función de la celda D18 no se incluye el parámetro ordenado ya que el mismo adopta el valor VERDADERO y se debe omitir.

Es importante notar que el valor VERDADERO en el parámetro ordenado nos permite buscar el porcentaje de descuento dentro de un rango. Al realizar una búsqueda que no es exacta, Excel nos devuelve el valor inmediato anterior al límite superior del rango correspondiente al valor_buscado. Veamos algunos ejemplos:

Valor_buscado	Rango		Porcentaje de descuento		Resultado
	Lím inferior	Lím superior	Lim inferior	Lim superior	
263,25	150	300	2%	3,50%	2%
95,40	0	150	0%	2,00%	0%
670,60	500	700	5%	7,50%	5%

Sintaxis de la función BUSCARH

=BUSCARH(Valor_buscado, Matriz buscar_en,Indicador_filas,ordenado)

Todos los argumentos se resuelven de igual modo que la función BUSCARV con la única diferencia que el Indicador_columnas debe cambiarse por Indicador filas.

Continuando con el ejemplo planteado, coloque formato porcentaje con 2 decimales a la celda D18.

Para calcular el importe de descuento, ubíquese en la celda E18 y realizó la multiplicación de las celdas E17*D18.

Calcule el importe total a pagar en la celda F19 como la resta entre el importe (celda E17) menos la bonificación (celda E18).



Ejercicios de autoevaluación

Ejercicio 1

En la celda G6 escriba como título COD-FDIT, en la celda H6 NOMBRE EDITORIAL, en la celda I6 AUTOR, en la J6 STOCK y en la K6 FECHA RECEPCION. Coloque ajuste de texto a todas las celdas de título y realice borde simple a toda la planilla.



Ejercicio 2

En la celda H7 escriba una función BUSCARH que le permita mostrar el nombre de la editorial a partir del código y según la tabla de editoriales que se encuentra en la Hoja 3.



Ejercicio 3

En la celda I7 escriba un función BUSCARV que muestre el autor de los libros a comprar teniendo en cuenta el código de la columna A; este dato se encuentra en la tabla LIBROS.



Ejercicio 4

En la celda J7 escriba una función BUSCARV que muestre el stock actual de cada código de libro; este dato se encuentra en la tabla LIBROS.



Ejercicio 5

En la celda K7 escriba una fórmula que muestre la fecha de recepción en que se dispondrán de los libros comprados. Tendrá que sumar a la fecha de compra programada (celda F4) la demora que tiene cada editorial en el envío de los libros, dato que debe obtener con un BUSCARH en la hoja 3 del libro a partir del código de editorial.



Ejercicio 6

Copie las fórmulas realizadas para todos los libros. Puede seleccionar el bloque H7:K7 y utilizando el botón de llenado automático completar las celdas hasta la fila 16.



Funciones Financieras

Una empresa realiza un depósito de \$1.000 en una entidad financiera, a retirar a 180 días de la fecha, a una tasa de interés de 1% efectiva para 30 días; se desea determinar cual será el monto que retirara del banco luego de transcurrido este tiempo.

Abra el archivo 0304.xls desde la carpeta Guías del CD.

Funciones para operaciones de capitalización o descuento y rentas

Microsoft Excel utiliza las mismas funciones para problemas de capitalización o descuento y de rentas, variando el empleo u omisión de los argumentos a los efectos de realizar uno u otro calculo. En este apartado se analizaran algunas de las funciones que consideramos importantes.

En la hoja Valor Futuro de este archivo se presentan los datos que surgen del problema anterior.

Un detalle a tener en cuenta es que en las funciones financieras, los importes que representan egresos de fondos deben indicarse como valores negativos, mientras que los que representan ingresos, deben ser positivos.

	A	B
4	Tasa:	1%
5	Cantidad de Períodos:	6
6	Valor Actual:	-1000
7		
8	Valor Futuro:	\$ 1.061,52
9		

De acuerdo con el planteo, se trata de una operación financiera de capitalización, donde lo que se pretende calcular es el valor futuro de la inversión. Para obtener el mismo, es de aplicación la función VF cuya sintaxis es la siguiente:

$VF(Tasa;Nper;Pago;Va;Tipo)$

Si se la utiliza para capitalización, deben omitirse los argumentos Pago y Tipo, u omitirse Tipo e indicar como valor de Pago cero.

En el ejemplo planteado, el valor actual del depósito bancario se ingresó como un número negativo puesto que representa una salida de dinero con destino al banco, mientras que el valor futuro que se obtendrá luego de aplicar la fórmula es un valor positivo, dado que representa la entrada de dinero a la empresa después de transcurridos los 6 períodos. A este valor Microsoft Excel le asigna automáticamente formato numérico monetario con dos decimales.

Nota: Si bien los valores podrían ingresarse a los argumentos de la función como constantes, el hecho de colocarlos como referencias a celdas permite realizar análisis de sensibilidad, modificando los distintos valores y observando el resultado de la inversión, ingresando tasas propuestas por diversos bancos, especulando con la cantidad de períodos a dejar depositado el dinero, modificando el depósito inicial, etc.

La función VF dispone de cinco argumentos que seguidamente describimos:

Tasa: Representa la tasa efectiva expresada para el período de tiempo en el cual capitaliza el valor actual o que corresponde a cada uno de los períodos de una renta. Siempre debe ingresarse a la función en tanto por uno, o bien seguida del signo %. Esta tasa debe estar expresada en función del tiempo de duración del período.

Nper: Representa el número total de períodos en el cual se somete a valoración el valor actual de dicha inversión, si el problema a resolver es de capitalización; o bien la cantidad total de períodos en los cuales se efectúan pagos para lograr el valor futuro de una renta.

Pago: Representa los pagos periódicos que se efectúan en una renta. En el caso de que el problema a resolver sea de capitalización, este argumento debe omitirse o debe indicarse cero.


Va: Representa el valor actual, es decir, el capital inicial en un problema de capitalización. Para el caso de rentas, tiene otro tratamiento que se explicara oportunamente.

Tipo: Se utiliza para indicar si los vencimientos de los pagos de una renta se producen al inicio o al final de cada período, debiéndose indicar 1 si la misma es de pagos adelantados o 0 si es de pagos vencidos. En el caso de que la operación financiera a resolver sea de capitalización o descuento, este argumento debe omitirse.

En la celda B8 escriba la siguiente fórmula o utilice el asistente de funciones para ingresar los argumentos de la misma:



=VF(B4;B5;;B6)



Nota: Al no indicarse Pago deben respetarse los separadores de argumentos, es decir, deben colocarse dos puntos y comas seguidos.

Si, en cambio, en lugar de tratarse de un depósito bancario, se tratara de un préstamo que solicita la empresa a una entidad financiera a devolver luego de 6 períodos, a una tasa periódica del 1 %, capitalizando periódicamente los intereses, el valor actual representaría el capital que recibiríamos de dicha entidad y por lo tanto ese valor debería ingresarse como un número positivo, mientras que el valor resultante de la fórmula sería un valor negativo que representaría el egreso que debería afrontar la empresa luego de transcurridos esos 6 períodos.

Siguiendo el mismo ejemplo, supongamos ahora que, siendo conocidos el valor futuro, la tasa de interés y la cantidad de períodos, deseamos determinar el valor actual de dicha inversión. Es decir, cual sería la suma de dinero que deberíamos depositar en una entidad financiera a una tasa periódica del 1% para 30 días, capitalizando los intereses periódicamente, para obtener al cabo de 6 períodos un valor futuro de \$1061,52.

Posiciónese en la hoja Valor Actual del libro de cálculo.

La función que resuelve esta operación es la función VA, cuya sintaxis es la siguiente:

VA(Tasa,Nper,Pago,Vf,Tipo)

	A	B
4	Tasa:	1%
5	Cantidad de Períodos:	6
6	Valor Futuro:	1061,520151
7		
8	Valor Actual:	-\$ 1.000,00
9		


La mayoría de estos argumentos ya fue explicada al abordar la función VF, la diferencia es que en aquella se indicaba como tercer argumento el valor actual y en ésta el valor futuro.

En este caso, el valor futuro debe ingresarse a la función como un número positivo, porque es el valor que se retirara del banco y, por lo tanto, constituye un ingreso a la empresa y representa el valor que se lograra luego de los Nperiodos capitalizando periódicamente el valor actual a la tasa periódica. Una vez ingresada la función, observara que el valor actual devuelto por la misma en formato monetario y con dos decimales es un número negativo puesto que representa el valor a depositar en el banco, y por lo tanto una salida de dinero de la empresa. No deben ingresarse a la función los argumentos pago y tipo.

En la celda B8 ingrese la siguiente fórmula o utilice el asistente para el diseño de misma:



=VA(B4;B5;;B6)



Microsoft Excel procederá a descontar el valor futuro utilizando la tasa indicada, en la cantidad de períodos señalados en el argumento Nper, es decir, 6 períodos. El descuento que realiza esta función es un descuento racional, puesto que para realizar esta operación financiera utiliza una tasa de interés y no una tasa de descuento.

Nota: En esta fórmula, a los efectos de la explicación, el valor futuro fue ingresado con 6 decimales. Tenga presente que en las operaciones financieras tanto el valor a depositar como el valor a retirar del banco estarán expresados en dos decimales por tratarse de cantidades monetarias.

Si ahora lo que se pretendiera lograr es la tasa de interés, siendo conocidos el valor actual, el valor futuro y la cantidad de períodos, deberíamos utilizar la función TASA. Esta función responde a la siguiente sintaxis:


TASA(Nper,Pago,Va,Vf,Tipo)

	A	B
4	Cantidad de Períodos:	6
5	Valor Actual:	-1000
6	Valor Futuro:	1061,520151
7		
8	Tasa:	1%
9		

Posiciónese en la hoja Tasa, donde encontrará los datos necesarios para hacer este cálculo e ingrese la siguiente fórmula o recurra al asistente:

=TASA(B4;;B5;B6)

El resultado obtenido por esta función es 0,01. es decir, la tasa esta expresada en tanto por uno. A los efectos de la visualización de la misma, Microsoft Excel le asigna formato porcentaje.



Esta tasa esta expresada para el tiempo que constituye el período de capitalización, vale decir, en el ejemplo propuesto es la tasa efectiva para 30 días.

Nota: Si en caso de tratarse de un depósito se hubiera tratado de un préstamo bancario, desde el punto de vista conceptual, el valor actual debería ingresarse a la función como un valor positivo y el valor futuro como negativo. No obstante el valor absoluto devuelto por la función sería idéntico.


Por último, lo que quedaría por analizar es el cálculo de la cantidad de períodos a la que debe someterse un cierto capital, a una tasa de interés conocida, a los efectos de lograr un determinado valor futuro.

	A	B
4	Tasa:	1%
5	Valor Actual:	-1000
6	Valor Futuro:	1061,520151
7		
8	Cantidad de Períodos:	6,000000038
9		

Siguiendo con el ejercicio propuesto, la pregunta sería: ¿durante cuantos períodos de tiempo deben depositarse en un banco \$1000 a la tasa efectiva del 1% para 30 días para lograr retirar \$1061,52?

La función a utilizar es NPER que responde a la siguiente sintaxis:

NPER(Tasa,Pago,Va,Vf,Tipo)



Posiciónese en la hoja Nperiodos donde encontrará los datos del problema y en la celda B8 escriba la siguiente fórmula o utilice el asistente para funciones:



=NPER(B4;;B5;B6)

Los \$1000 para convertirse en 1061,52 invertidos a una tasa efectiva del 1% para 30 días, deberán dejarse depositados durante 6 periodos de 30 días.

Nota: Los periodos obtenidos por esta función son siempre constantes y tienen una duración, cada uno de ellos, equivalente al tiempo en el cual está expresada la tasa, y en el cual capitalizan los intereses.

Veamos ahora cómo sería la aplicación de las funciones desarrolladas anteriormente en las operaciones financieras de rentas:

Supongamos que durante diez periodos de 30 días se depositan periódicamente en un banco \$100 a una tasa del 1% para 30 días, capitalizando los intereses periódicamente, ¿cual sería el valor futuro de esa renta al cabo de los 300 días? Dado de que se trata de depósitos en un banco, se asume que los pagos vencen al inicio de cada período, es decir es una renta de pagos adelantados.

Los argumentos de esta función ya fueron explicados oportunamente.

Posiciónese en la hoja Valor Futuro Rentas del archivo.


Observe que aparece aquí un valor para el argumento Pago, siendo el mismo de —100 (como número negativo, puesto que significa una erogación). Además, aparece también un valor para el argumento tipo, siendo 1 en este caso puesto que los pagos se efectúan al principio del período. Si los pagos, en cambio, fueran al final del período, el valor de este argumento debería ser 0 u omitirse.

	A	B
4	Tasa:	1%
5	Cantidad de Periodos:	10
6	Pago Periódico:	-100
7	Tipo:	1
8		
9	Valor Futuro:	\$ 1.056,68
10		

Para poder aplicar esta función, los pagos tienen que ser todos iguales, y el tiempo de duración de cada período debe ser constante. A su vez la tasa de interés no debe variar a lo largo de todo el tiempo de valoración.

En la celda B8 escriba la siguiente fórmula o emplee el asistente:

$$=VF(B4; B5; B6;; B7)$$



Nota: Observe que se omitió indicar el cuarto argumento, es decir Va (valor actual), puesto que en el caso planteado sólo se realizan pagos periódicos. Si, en cambio, además de los pagos periódicos, en el momento inicial se realizara también el depósito por única vez de otra suma de dinero, este valor debería ingresarse en Va y estaría sujeto a interés durante los Nperíodos. Suponga entonces en el caso planteado que además de los \$100 que se depositan al inicio del primer período, se deposita también en ese momento \$1000, ¿cuál sería el valor futuro obtenido? La fórmula que resuelve esto es la siguiente: =VF(1%;10;-100;- 1000;1), obteniendo como resultado: 2161,3056, es decir =VF(1%; 10;-100;;1) + VF(1%;10;;-1000).

Merece desarrollarse, por no haber sido encarada anteriormente, la función Pago que permite encontrar el valor de la cuota periódica que se paga en una renta, a los efectos de construir un valor futuro, capitalizando los intereses periódicamente a una determinada tasa.

	A	B
4	Tasa:	1%
5	Cantidad de Períodos:	10
6	Valor Futuro:	1056,68347
7	Tipo:	1
8		
9	Pago Periódico:	-\$ 100,00
10		

Posiciónese en la hoja Pago.


Siguiendo con el mismo ejemplo, nos proponemos determinar cual sería el importe a depositar cada 30 días a una tasa efectiva del 1% para 30 días durante 10 períodos para lograr un valor futuro de \$1.056,68. Para lograr esto, debemos aplicar la función Pago, cuya sintaxis es:

$$=PAGO(Tasa,Nper,Va,Vf,Tipo)$$

En la celda B8 ingrese la siguiente expresión o recurra al asistente para funciones:



=PAGO(B4;B5;;B6;B7,



El resultado de esta fórmula nos indica que deben depositarse cada 30 días \$100.

El mismo resultado pudo haberse obtenido indicando en la función Pago el valor actual de la renta, en lugar del valor futuro. Para ello, previamente tendríamos que disponer del valor actual, el cual es 956,601757.

Si ahora volvemos a calcular la cuota a depositar pero en función del valor actual, tendremos: =PAGO(1%;10;956,601757;;1) y obtendremos que igualmente el valor a depositar es de \$100.

Nota: Dentro de la función Pago debe indicarse el valor actual o el valor futuro de la renta, pero no ambos, puesto que el valor de la cuota que se obtendría no sería el correcto. Sólo sería justificable indicar ambos argumentos, para el caso ya explicado en notas anteriores, cuando además de las cuotas periódicas se depositara en el momento inicial y por única vez un valor capital distinto. Entonces, en este caso sería obligatorio como tercer argumento de la función indicar dicho importe y como cuarto argumento el valor futuro. En el ejemplo a que hacemos referencia, además de los 10 pagos de \$100 que se realizaban al inicio de cada período, al principio de primer período de valoración se depositaban también \$1000. La fórmula que resuelve el valor del pago sería entonces: =PAGO(1%;10;-1000; 2161,3056;1), cuyo resultado es \$100. En la fórmula planteada, el valor de tercer argumento, es decir -1000, desde el punto de vista financiero, no representa conceptualmente el valor actual de la renta, sino simplemente ese capital distinto del de las cuotas o pagos y que se deposita al inicio del tiempo total de valoración por única vez.

También es posible, a partir del valor actual o el valor futuro de la renta, la cantidad de períodos y el pago, encontrar la tasa a la que se valora el conjunto de capitales. La fórmula que permitiría encontrar la tasa en el caso propuesto sería la siguiente:


=TASA(10;-100;;1056,68347;1), o bien =TASA(10;400;956,601757;;1), según se indique el valor futuro, o el valor actual de la renta como argumento de la función.

Si se desea conocer la cantidad de períodos a que se deben someter los \$100 que se depositan periódicamente a la tasa del 1% para 30 a los efectos de lograr un valor futuro de \$1056,68, o un valor actual de esos flujos de fondos de \$956,60, la fórmula sería la siguiente:


=NPER(1 %;-100;;1056,68347;1) o bien =NPER(1 %;-100;956,601757;;1), según se disponga del valor futuro o el valor actual de la renta.

Por último, resulta conveniente realizar un comentario final respecto del quinto argumento de todas estas funciones, es decir Tipo. Si los pagos que se efectúan vencen al principio de cada período, debe indicarse 1 como valor para este argumento y sería el caso del ejemplo propuesto en este material para explicar las funciones bajo análisis. Pero, si en lugar de tratarse de importes que se depositan periódicamente en un banco, se tratara de 10 documentos de \$100 cada uno y el primero de ellos vence a los 30 días de la fecha, el segundo a los 60 y así sucesivamente a intervalos regulares de 30 días, y concurríamos en el día de hoy a descontarlos en un banco, ¿cual sería el valor actual de esos documentos al primer período, siendo que el primer pago vence al final del mismo?

En este caso aplicaríamos la función VA ya explicada, indicando dentro de la misma como quinto argumento 0, de lo que surge la expresión que se representa en la siguiente imagen, cuyo resultado es \$947,13. Observe que este importe es inferior al obtenido al calcular el valor actual de los depósitos bancarios, puesto que como los pagos se efectúan al final de cada período, al calcular el valor actual estos flujos de fondos están sujetos a 30 días mas de descuento.



Por consiguiente, si calculamos el valor futuro de los mismos, obtendremos un importe menor al logrado en el caso de los depósitos bancarios, puesto que los distintos pagos están sujetos a un período menos de capitalización, siendo dicho valor \$1046,22.



Valor neto actual y tasa interna de retorno

Entiéndese por Valor Neto Actual al valor que tienen los flujos futuros de caja de una inversión, descontados al momento inicial, a una determinada tasa. A esta tasa se la denomina tasa requerida del proyecto.

Para la explicación de esta función recurriremos a un ejemplo. Supongamos que se dispone de un capital de \$100.000 y se tienen tres alternativas de inversión.

Alternativa 1: Invertir los \$100.000 en la compra de títulos públicos, estimándose que 25% de los mismos pueden ser rescatados al año de realizada la compra, por un valor de \$25.000, al año siguiente otro 25% a \$30.000, y en los dos siguientes años, \$35.000 y \$40.000 respectivamente, que constituyen el 50% restante.

Alternativa 2: Realizar un préstamo que será devuelto de la siguiente forma, \$10.000 en forma adelantada al momento del otorgamiento del mismo, y el 90% restante en 8 cuotas que se cobrarán semestralmente de \$10.000, \$12.000, \$14.000, \$16.000, \$18.000, \$20.000, \$22.000, \$24.000, con vencimiento al final de cada uno de los semestres.

Alternativa 3: Una inversión a más largo plazo, que consiste en la constitución de una empresa para lo cual se aportaría al momento de la constitución, un capital de \$80.000, estimándose que el primer año no generara utilidades pero demandara un refuerzo de capital de \$20.000, para luego arrojar, al final del ejercicio económico del segundo año, y durante 5 años \$18.000, \$26.000, \$35.000, \$45.000 y \$58.000, en concepto de utilidades e ingresos provenientes de la ventas de las acciones de la compañía, puesto que además de la ganancia, se pretende recuperar el capital invertido en ese tiempo.

Se desea averiguar cual es el proyecto que genera una mayor rentabilidad logrando devolver el capital invertido, teniendo en cuenta que otra posibilidad consistiría en el depósito de ese dinero a plazo fijo, obteniéndose una tasa libre de riesgos del 6% para 180 días y que se toma como referencia.

Para resolver esta situación Microsoft Excel nos propone dos funciones, una de ellas es VNA que pasaremos a explicar seguidamente, y otra es TIR que encararemos con posterioridad.

	A	B	C	D
4		Alternativa 1	Alternativa 2	Alternativa 3
5		-100000	-90000	-80000
6	1º Semestre	0	10000	0
7	2º Semestre	25000	12000	-20000
8	3º Semestre	0	14000	0
9	4º Semestre	30000	16000	18000
10	5º Semestre	0	18000	0
11	6º Semestre	35000	20000	26000
12	7º Semestre	0	22000	0
13	8º Semestre	40000	24000	35000
14	9º Semestre			0
15	10º Semestre			45000
16	11º Semestre			0
17	12º Semestre			58000
18				
19	VNA:	-\$ 3.978,46	\$ 11.114,24	\$ 10.092,60
20				

Ubíquese en la hoja VNA y TIR, en la cual encontrará los datos correspondientes a los flujos de caja.

En las distintas columnas de esta planilla fueron cargados los datos de cada una de las tres alternativas que surgen del planteo. A efectos de poder utilizar la misma tasa para descontar los flujos de caja y dado que en la alternativa 2 se realizan movimientos semestrales de fondos, los flujos de fondos de todas las inversiones fueron organizados en la planilla en forma semestral. A los efectos de poder realizar la comparación, los períodos de tiempo deben ser iguales entre las distintas alternativas de inversión. Por otro lado, VNA exige que los períodos de tiempo dentro de un mismo proyecto de inversión sean todos iguales, pudiendo ser distintos los importes de los flujos en cada período, valores positivos o negativos que representarían ingresos y egresos respectivamente; al menos uno de los valores debe ser negativo.

Nota: Los valores de la fila 1º Semestre corresponden a los movimientos de fondos al final del 1º semestre, los de la fila 2º Semestre al final de ese semestre y así sucesivamente.

La función VNA descuenta, con la tasa indicada, los flujos de fondos positivos y

negativos al momento inicial de valoración y luego suma estos flujos descontados. La sintaxis de esta función es la siguiente:


VNA(Tasa; Valor1; Valor2;....)

Donde tasa es la tasa requerida para el proyecto y los siguientes argumentos constituyen la lista de flujos positivos y negativos que se descuentan al momento inicial, utilizando dicha tasa. Estos valores pueden indicarse como constantes dentro de la función o también ingresarse en una matriz de una sola columna, de una sola fila, o bien de varias filas y columnas.

En B19 escriba la siguiente fórmula para la primera alternativa de inversión, y luego cópiela en las celdas C19 y D19 para las restantes alternativas:



=VNA(6%;B5:B17)



Si bien en las alternativas de inversión 1 y 2 el rango de datos abarca de a fila 5 hasta la fila 13 inclusive, el rango de la matriz ingresado en la función incluye hasta la fila 17 a efectos de copiarlo en las otras columnas, pero no influye sobre el resultado alcanzado.

Obsérvese que la alternativa 1 y la alternativa 3 al final del primer semestre no ofrecen movimiento de fondo alguno, por lo tanto, a los efectos del buen funcionamiento de la función, es necesario colocar cero en dichos momentos, y así análogamente en todos aquellos semestres donde no se verifiquen flujos de fondos.

En la alternativa de inversión número 2, el flujo inicial es de —\$90.000 puesto que representa el egreso neto, dado que se prestan \$100.000, pero al mismo tiempo se percibe en forma adelantada \$10.000. En la inversión número 3, existen dos flujos negativos, puesto que esta alternativa demanda una inversión inicial de \$80.000 y luego, al final del primer año, un nuevo aporte de \$20.000.

Analizando los resultados obtenidos se observa que la alternativa de inversión 1 no alcanza a cubrir la tasa requerida de rendimiento del proyecto, por lo que faltan \$3.978,46 para alcanzar la utilidad que se obtendría de depositar ese dinero a plazo fijo al 6% semestral, por lo que es descartada como alternativa viable. Las

alternativas 2 y 3 generan excedentes sobre la tasa pretendida y que se tomó como referencia. De ambas, la más conveniente es la alternativa 2 puesto que genera un mayor excedente en un menor tiempo.

El valor que nos devuelva la función VNA, aplicado a la segunda alternativa, nos indica que esta inversión nos devuelve el capital invertido, nos retribuye con una tasa semestral del 6%, y además nos genera \$11.114,24 de excedente.

Otra herramienta que podríamos aplicar para evaluar la inversión es la TIR (Tasa Interna de Retorno). Ésta es la tasa a la cual, descontados los flujos positivos y negativos, es decir el conjunto de ingresos y egresos realizados a intervalos regulares, arroja un valor neto actual nulo.

La distribución de los flujos de fondos en la planilla sería igual al analizado para el VNA.

La sintaxis de la función es la siguiente:

TIR(Valores,Estimar)


El primer argumento de la función corresponde al bloque o rango de la hoja donde se ha ingresado la matriz de los valores del flujo de caja; mientras que el segundo argumento, que es de carácter opcional, corresponde a la tasa que el decisor considera que podría ser la Tasa Interna de Retorno y a partir de la cual, Microsoft Excel comienza a interpolar para alcanzar la tasa correcta. La primera tasa de interpolación será la indicada en el argumento estimar o si éste se omitiere será 0,1 es decir una tasa del 10%; a partir de allí Excel comienza a iterar hasta alcanzar la tasa que anule los flujos de fondos descontados, teniendo una exactitud del 0,00001%. En el caso de no llegar a un resultado después de 20 iteraciones, esta función devolverá el error #¡NUM!

En B21 escriba la siguiente fórmula y luego cópiela en C21 y D21:

=TIR(B5:B17;6%)



En el caso planteado se partió del 6%, porque era la tasa de referencia con la cual se comparaban las distintas alternativas de inversión, pero podría haberse indicado cualquier otra tasa razonable, incluso podría haberse omitido este argumento. En el caso de omitirse el argumento a estimar Microsoft Excel asumirá como primer valor de interpolación el 0,1 es decir una tasa periódica del 10%.



Es de destacar que esta herramienta de cálculo es aplicable para sólo a aquellas inversiones donde los ingresos y egresos se distribuyen en períodos iguales, independientemente de la amplitud de período, y la TIR lograda estaría expresada como tasa para ese período.

Analizando los resultados llegaremos a conclusiones similares a las obtenidas de aplicar la función VNA, es decir que la alternativa 1 no alcanza a cubrir la tasa del 6% que se tomó como referencia, siendo la tasa que rinde esta inversión del 5,11% por semestre. De las restantes alternativas, desde el punto de vista de la rentabilidad, la: mas conveniente es la inversión 2. Esta inversión logra devolvemos el capital invertido y nos paga un interés del 8,77% semestral.

	A	B	C	D
4		Alternativa 1	Alternativa 2	Alternativa 3
5		-100000	-90000	-80000
6	1º Semestre	0	10000	0
7	2º Semestre	25000	12000	-20000
8	3º Semestre	0	14000	0
9	4º Semestre	30000	16000	18000
10	5º Semestre	0	18000	0
11	6º Semestre	35000	20000	26000
12	7º Semestre	0	22000	0
13	8º Semestre	40000	24000	35000
14	9º Semestre			0
15	10º Semestre			45000
16	11º Semestre			0
17	12º Semestre			58000
18				
19	VNA:	-\$ 3.978,46	\$ 11.114,24	\$ 10.092,60
20				
21	TIR:	5,11%	8,77%	7,34%
22				

Para concluir, debemos tener en cuenta algunas cuestiones respecto de la función TIR:

- El primer argumento debe contener por lo menos un valor positivo y un valor negativo para poder realizar el cálculo.

- La función TIR, a los efectos de descontar los flujos de caja, se basa en el orden en que se ubican los valores dentro del rango referenciado como primer argumento de la función. Para obtener el resultado correcto, el orden de los valores debe corresponderse con el momento en que se efectivizan los ingresos y egresos.
- Si en algún período no se verifican ingresos o egresos, como valor del mismo es necesario que se indique cero.
- Si el resultado obtenido es el valor de error #¡NUM!, o si el valor obtenido no se aproximara a la tasa estimada, debería procederse a un nuevo intento indicando como segundo argumento un valor diferente del anterior.

Ejercicios de autoevaluación

Ejercicio 1

Se solicita un préstamo a un banco de \$2.500 a devolver en un solo pago a los 120 días de la fecha, a una tasa del 1,5% para 30 días, capitalizando los intereses cada 30 días. Determine el valor futuro de esta operación financiera.



Ejercicio 2

El valor futuro de un depósito a plazo fijo es de \$3.036. Determinar cuanto dinero se depositó, si la operación fue a 60 días a una tasa de interés del 2,5% para 60 días.



Ejercicio 3

¿Cuántos períodos se debe dejar depositados \$2.000 en un banco, a la tasa del 1,25% para 30 días, para retirar \$2.287,01, capitalizando los intereses cada 30 días?



Ejercicio 4

Se recibe de un banco en calidad de préstamo \$2.000, debiéndose devolver a los 180 días \$2.153. ¿Cual es la tasa periódica que aplica el banco, si los intereses capitalizan cada 45 días?



Ejercicio 5

Se abre una caja de ahorros en un banco, depositándose en el momento



de apertura \$800. Cada 60 días se depositan en dicha cuenta \$150, depositándose el primer pago a los 60 días de la apertura de la cuenta, el segundo a los 120 días y así sucesivamente durante un año, capitalizando los intereses cada 60 días a la tasa del 1,9% para 60 días. Determine cual será el valor futuro obtenido luego de finalizado el sexto período.

Ejercicio 6

Una empresa esta planificando comprar una determinada maquinaria. Pagara la misma en 18 cuotas de \$120 cada una y de acuerdo con lo pactado con el vendedor la tasa de interés para la operación es de 1,4% para 30 días. Los pagos se efectúan cada 30 días, realizándose el primer en el momento de la compra. Se desea determinar cual sería el precio de contado de esa maquinaria, asumiendo que el mismo es el valor actual de las cuotas, descontadas a la tasa convenida.



Ejercicio 7

Un proyecto de inversión exige un desembolso \$30.000 en el primer año, y otro desembolso de \$10.000 en el tercer año, generando ingresos a partir del sexto año durante 10 períodos consecutivos de 10 años, de \$10.000. Se desea determinar si la inversión es rentable, siendo que la tasa representativa del rendimiento esperado es del 16% anual. Obtenga el VNA y la TIR.



Validación de Datos

Por validar, se entiende la posibilidad de controlar el ingreso de un cierto tipo de datos a una celda o un rango de celdas, o el intervalo dentro del cual se debe encontrar el valor a ingresar. De este modo se evita el ingreso de datos que pueden ser incorrectos. Por ejemplo, cuando ingreso la fecha de una factura, puedo validar que esta fecha se encuentre en un intervalo de fechas coherentes (p.e. entre el 1/1/2001 y el 31/12/2001), de modo que si se ingresa una fecha fuera de ese intervalo, me indique como erróneo dicho valor.

Abra el archivo 0305.xls, del CD, carpeta Guías en el cual tenemos un listado de facturas correspondientes a un grupo de cliente.

Es necesario ingresar la fecha correspondiente a cada factura, en la columna Fecha, restringiendo el ingreso para las fechas comprendidas dentro del año 2001.

Seleccione el rango D8:D20, luego el menú Datos / Validación, elija la ficha Configuración. En Permitir elija tipo de Datos Fecha u hora, y luego en Fecha Inicial y Fecha Final colocar los límites correspondientes (c sea entre 1/1/2001 y 31/12/2001) y haga clic en Aceptar.



Validación de datos

Configuración | Mensaje entrante | Mensaje de error

Criterio de validación

Permitir:
Fecha

Omitir blancos

Datos:
entre


Fecha inicial:
01/01/2001

Fecha final:
01/12/2001

Aplicar estos cambios a otras celdas con la misma configuración

Borrar todos | Aceptar | Cancelar

Ingrese en D8, intencionalmente, la fecha 5/10/2002, que es un valor fuera del límite, lo cual hará aparecer el siguiente Mensaje de error.



Si hacemos clic en el botón Reintentar, volveremos al modo edición de la celda para poder reingresar el valor. En caso de hacer clic en Cancelar, se anula directamente esta acción volviendo a la celda activa de la hoja.

Haga clic en Cancelar.

Podemos personalizar este Mensaje de error, para que nos muestre la leyenda que nosotros queramos, de modo que el usuario sepa el motivo de su error



Seleccione nuevamente el rango D8:D20. Luego en el menú Datos / Validación, elija la ficha Mensaje de error. Escriba como mensaje de error, Ingrese una fecha dentro del año 2001. Haga clic en Aceptar.

Vuelva a ingresar la fecha 5/10/2002 y vea como se muestra ahora el mensaje de error personalizado.

También es posible agregar Mensajes Entrantes, de modo que al posicionarse sobre esa celda, indique un mensaje advirtiendo al usuario sobre los datos a ingresar, de modo de evitar valores erróneos.

Seleccione una vez más el rango D8:D20. Luego en el menú Datos Validación, elija la ficha Mensaje Entrante. Escriba como mensaje entrante, Ingrese una fecha (sólo del año 2001). Haga clic en Aceptar.

Se puede observar ahora que al posicionarse sobre las celdas validadas, aparece el mensaje entrante.

Si deseamos permitir que la celda que se está restringiendo esté en blanco o si desea definir límites que utilicen una referencia de celda, o una fórmula que dependa de celdas que inicialmente están en blanco, deberemos comprobar que la casilla de verificación Omitir blancos esté activada.

Para hacer obligatorias las restricciones definidas en las celdas en blanco, tratándolas como si tuvieran ceros, desactive la casilla de verificación Omitir blancos.

Ejercicio de autoevaluación

Ejercicio 1:

En la columna Pagos del libro 0305.xls, es necesario ingresar los valores abonados por el cliente. Dichos Pagos sólo pueden ser positivos (mayores a 0) y no pueden superar al valor de la columna Importes, que representa el valor comprado.

Realice la correspondiente validación de las celdas del rango F8:F20, limitando el ingreso de valores a aquellos que se encuentren dentro del intervalo comprendido entre 0 y el valor correspondiente. Como Mensaje Entrante debe aparecer: "Ingrese el monto abonado", y como Mensaje de Error: "Solo valores positivos. No supere Importe." 

Nota: Aquí debemos validar primero sólo la celda F8, para valores entre 0 y la celda D8, y luego copiar F8 al resto de las celdas. Al copiar esta celda hacia abajo, por existir una referencia relativa de celda, hará que los valores correspondientes al Importe (D9, D10, etc.) aparezcan correctamente validados para las celdas correspondientes (es decir se copian los valores de validación).

Trabajo práctico 3

La librería CONTINENTAL cuyo canal de comercialización principal es el mayorista necesita automatizar la facturación que realiza a sus clientes. Para ello tiene un listado de productos, un listado de clientes, un cuadro de porcentajes y un modelo de factura.

1. Abra el archivo 03.xls de la carpeta Prácticos del CD.
2. Cambie el nombre a las hojas 1, 2, 3, 4 y 5 por Facturas, Clientes, Libros, Porcentajes y Análisis Financiero respectivamente.

En la hoja Libros.

3. En la celda E4 calcule el Precio de Venta incrementando el precio de costo con el porcentaje indicado en la celda F 1 . Redondee el resultado hacia arriba en 2 decimales.
4. Utilizando el controlador de relleno (Ver Capítulo 2, Referencia Absolutas y Relativas) copie a las celdas adyacentes la fórmula recién obtenida.
5. Seleccione el rango de celdas A4:E22. Coloque como nombre del bloque LIBROS (Ver Capítulo 2, Referencias Absolutas y Relativas).

En la hoja Facturas.

6. Ingrese el número 10 en la celda C1.
7. En la celda C2, escriba una función por la cual Microsoft Excel, a través del Nro. de Cliente muestre el Apellido del mismo.
8. En la celdas C3, C4 y C5 escriba las funciones para mostrar los datos del cliente ingresado en el ítem a (Nombre, Domicilio y Localidad).
9. En la celda C9 escriba una función para mostrar el Título del Libro teniendo en cuenta la columna código(celda A9). El resultado de dicha función dará #N/A (Ver Capítulo 3, Funciones de Búsqueda y Referencia) puesto que todavía no ingresamos el código.
10. En la celda A9 ingrese el número 15 (aparecerá la descripción) y en la celda B9 el número 1.
11. En la celda D9 escriba una función para mostrar el Precio de Venta teniendo en cuenta la columna Código.

12. En la celda E9 calcule el Importe multiplicando la cantidad por el Precio de Venta.

13. Para mejorar el aspecto de la planilla incorpore en la función de la celda C9 una función lógica que evalúe si está cargado el código del libro. De no estarlo debe mostrar espacios en blanco.

14. Complete los datos de las columnas Código y Cantidad ingresando los valores de la tabla.

CODIGO	CANTIDAD
15	1
17	1
9	1
7	2
11	1
4	2
1	1
5	3
3	1
2	1
10	4
12	2

15. Seleccione el bloque C9:E9 y copie las fórmulas para el resto de las filas.

16. En la celda E21 utilice el botón Autosuma para obtener el subtotal.

17. En la celda D22 debe mostrar el porcentaje de descuento a aplicar según la tabla de la hoja porcentajes y el subtotal calculado en el punto anterior. Aplique formato Porcentajes con 2 decimales.

18. En la celda E22 obtenga el importe del descuento.

19. En la celda E23 obtenga el Total de la factura restando al Subtotal el importe del Descuento.

20. En la celda E1 ingrese una función que tome la fecha actual del sistema.

21. Existen dos formas de pago, Contado y C. Corriente. En la celda E2 escriba alguna de ellas por ejemplo Contado.

22. En la celda E4 calcule la primer fecha de vencimiento teniendo que es 15 días posterior a la fecha ingresado (celda E1) si la forma de pago es C. Corriente caso contrario mostrar espacios en blanco.

23. En la celda E5 calcule la segunda fecha de vencimiento teniendo que es 30 días posterior a la fecha ingresado (celda E1) si la forma de pago es C. Corriente caso contrario mostrar espacios en blanco.

24. Aplique los formatos numéricos correspondientes según se muestran en el modelo adjunto.

Modelo Final Hoja Facturas

	A	B	C	D	E
1		Cliente Nro	10	Fecha	14/06/01
2		Apellido	GOMEZ	Forma de Pago	Contado
3		Nombre	JAVIER		
4		Domicilio	RODRIGUEZ 1733	1er Vencimiento	
5		Localidad	CORONDA	2do Vencimiento	
6					
7					
8	Código	Cantidad	Título del Libro	Precio Venta	Importe
9	15	1	Madame Pinnel	\$ 24,70	\$ 24,70
10	17	1	Moby Dick	\$ 18,18	\$ 18,18
11	9	1	La máquina del tiempo	\$ 5,03	\$ 5,03
12	7	2	Historias de Marilú	\$ 6,13	\$ 12,26
13	11	1	La odisea	\$ 8,34	\$ 8,34
14	4	2	El vendedor más grande	\$ 15,78	\$ 31,56
15	1	1	Alfa Spider el carro ideal	\$ 16,98	\$ 16,98
16	5	3	Guerra y paz	\$ 16,19	\$ 48,57
17	3	1	Cien años de soledad	\$ 8,75	\$ 8,75
18	2	1	Así habló Zaratustra	\$ 11,00	\$ 11,00
19	10	4	La mujer orquesta	\$ 7,48	\$ 29,92
20	12	2	La república	\$ 6,33	\$ 12,66
21			Subtotal		\$ 227,95
22			Descuento	18,00%	\$ 41,03
23			Total		\$ 186,92

En la hoja Análisis Financiero.

25. La empresa está analizando la posibilidad de invertir en la compra de una partida de libros \$50.000, entregando la mitad de esa suma en el momento de la compra, \$15.000 a los 30 días y los \$10.000 restantes a los 60 días. De la proyección de ventas que realizó la empresa, según las tratativas realizadas con determinados clientes, se estima que la totalidad de los libros serán vendidas dentro de los 270 días de la compra, generando ingresos recién a los 60 días de la adquisición por el valor de \$6.000, los cuales irán en crecimiento, previéndose valores de ventas a partir de allí, para cada período de 30 días de \$6.800, \$7.900, \$9.100, \$9.800, \$10.500, para luego disminuir a los 240 días y 270 días como consecuencia del agotamiento del stock, a \$3.500 y \$1.400 respectivamente. La empresa desea recuperar el capital en ese tiempo, y además obtener como mínimo una ganancia porcentual del 2% para 30 días. En la hoja Análisis Financiero del libro de cálculo realice un análisis utilizando las funciones VNA y TIR para determinar si la inversión permite recuperar el dinero invertido y alcanza la rentabilidad pretendida.

26. Si desea conservar este archivo resuelto guárdelo en su disco rígido.